

BNF DA LINGUAGEM DE PROGRAMAÇÃO LINGPON 1.0

```
1 <program> ::= <fbes> <inst> <strategy> <rules> <main>
2 | <fbes> <inst> <strategy> <rules>
3
4 <fbes> ::= <fbe>
5 | <fbe> <fbes>
6
7 <fbe> ::= FBE <id> <fbe_body> END_FBE
8
9 <inst> ::= INST <declarations> END_INST
10
11 <strategy> ::= STRATEGY <strategy_declaration> END_STRATEGY
12
13 <strategy_declaration> ::= NO_ONE
14 | BREADTH
15 | DEPTH
16
17 <declarations> ::= <declaration>
18 | <declaration> <declarations>
19
20 <declaration> ::= <type> <ids>
21
22 <ids> ::= <id>
23 | <id> COMMA <ids>
24
25 <rules> ::= <rule>
26 | <rule> <rules>
27
28 <rule> ::= RULE <rule_body> END_RULE
29 | RULE <id> <rule_body> END_RULE
30
31 <rule_body> ::= <decl_condition> <decl_action>
32 | <decl_properties> <decl_condition> <decl_action>
33
34 <decl_properties> ::= PROPERTIES <properties_body> END_PROPERTIES
35
36 <properties_body> ::= <properties_type> <value>
37
38 <properties_type> ::= PRIORITY
39 | KEEPER
40
41 <decl_condition> ::= CONDITION <condition_body> END_CONDITION
42 | CONDITION <id> <condition_body> END_CONDITION
43
44 <condition_body> ::= <subcondition> <operator> <condition_body>
45 | <subcondition>
46
47 <operator> ::= AND
48 | OR
49
50 <subcondition> ::= SUBCONDITION <id> <subcondition_body>
51 END_SUBCONDITION
52
53 <subcondition_body> ::= <premise> AND <subcondition_body>
54 | <premise>
55
56 <premise> ::= PREMISE <exp>
57 | PREMISE <id> <exp>
58 | PREMISE IMP <exp>
```

```

59 | PREMISE IMP <id> <exp>
60
61 <exp> ::= <fator> <comp> <fator>
62
63 <comp> ::= EQ
64 | NE
65 | LT
66 | GT
67 | LE
68 | GE
69
70 <fator> ::= <id>
71 | NUMBER
72 | boolean
73 | FLOATVALUE
74 | STRINGVALUE
75 | CHARVALUE
76
77 <boolean> ::= TRUE
78 | FALSE
79
80 <decl_action> ::= ACTION <action_body> END_ACTION
81 | ACTION <id> <action_body> END_ACTION
82
83 <action_body> ::= <action_elements> <action_body>
84 | <action_elements>
85
86 <action_elements> ::= <instigation>
87 | <method_use>
88 | <exp> SEMICOLON
89
90 <instigation> ::= INSTIGATION <method_use>
91 | INSTIGATION <id> <method_use>
92
93 <method_use> ::= <id> LP RP SEMICOLON
94
95 <id> ::= ID
96 | ID POINT <id>
97
98 <decl_attributes> ::= ATTRIBUTES <attributes> END_ATTRIBUTES
99
100 <attributes> ::= <attributes_body>
101 | <attributes_body> <attributes>
102
103 <attributes_body> ::= <type> <id> <value>
104 | <type> <id> SEMICOLON
105
106 <type> ::= BOOLEAN
107 | INTEGER
108 | PFLOAT
109 | STRING
110 | CHAR
111
112 <value> ::= NUMBER
113 | FLOATVALUE
114 | STRINGVALUE
115 | CHARVALUE
116 | <boolean>
117
118 <decl_methods> ::= METHODS <methods> END_METHODS
119

```

```

120 <methods> ::= <method_body>
121           | <method_body> <methods>
122
123 <method_body> ::= METHOD <id> LP <id> ASSIGN <id>
124                <method_operator> <value> RP
125           | METHOD <id> LP <id> ASSIGN <id>
126                <method_operator> <id> RP
127           | METHOD <id> LP <id> ASSIGN <value> RP
128           | METHOD <id> LP <id> ASSIGN <id> RP
129           | METHOD <id> LP <id> RP INNER_CODE_METHOD
130           | METHOD <id> LP RP INNER_CODE_METHOD
131
132 <method_operator> ::= PLUS
133                   | MINUS
134                   | MULT
135                   | DIV
136
137 <main> ::= MAIN INNER_CODE_MAIN

```

Código 1: BNF da linguagem de programação LingPON 1.0
Fonte: Adaptado de Ferreira, 2015